(eval-when (:compile-toplevel :load-toplevel :execute)
  (sb-int:defconstant-eqx OPT
    #+swank '(optimize (speed 3) (safety 2))
    #-swank '(optimize (speed 3) (safety 0) (debug 0))
    #'equal)
  #+swank (ql:quickload '(:cl-debug-print :fiveam) :silent t)
  #-swank (set-dispatch-macro-character
           ;; enclose the form with VALUES to avoid being captured by LOOP macro
           #\# #\> (lambda (s c p) (declare (ignore c p)) `(values ,(read s nil nil t)))))
#+swank (cl-syntax:use-syntax cl-debug-print:debug-print-syntax)
#-swank (disable-debugger) ; for CS Academy

;; BEGIN_INSERTED_CONTENTS
(declaim (inline read-line-into))
(defun read-line-into (buffer-string &key (in *standard-input*) (term-char #\Space))
  "Receives ASCII inputs and returns multiple values: the string and the end
position.

This function calls READ-BYTE to read characters though it calls READ-CHAR
instead on SLIME because SLIME's IO is not bivalent."
  (declare (inline read-byte)) ; declaring (sb-kernel:ansi-stream in) will be faster
  (loop for c of-type base-char =
           #-swank (code-char (read-byte in nil #.(char-code #\Newline)))
           #+swank (read-char in nil #\Newline)
        for idx from 0
        until (char= c #\Newline)
        do (setf (char buffer-string idx) c)
        finally (when (< idx (length buffer-string))
                  (setf (char buffer-string idx) term-char))
                (return (values buffer-string idx))))

(defmacro with-buffered-stdout (&body body)
  "Buffers all outputs to *STANDARD-OUTPUT* in BODY and flushes them to
*STANDARD-OUTPUT* after BODY has been done (without error). Note that only
BASE-CHAR is allowed."
  (let ((out (gensym)))
    `(let ((,out (make-string-output-stream :element-type 'base-char)))
       (let ((*standard-output* ,out))
         ,@body)
       (write-string (get-output-stream-string ,out)))))

(defmacro dbg (&rest forms)
  #+swank
  (if (= (length forms) 1)
      `(format *error-output* "~A => ~A~%" ',(car forms) ,(car forms))
      `(format *error-output* "~A => ~A~%" ',forms `(,,@forms)))
  #-swank (declare (ignore forms)))

(defmacro define-int-types (&rest bits)
  `(progn
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "UINT~A" b)) () '(unsigned-byte ,b))) bits)
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "INT~A" b)) () '(signed-byte ,b))) bits)))
(define-int-types 2 4 7 8 15 16 31 32 62 63 64)

(declaim (inline println))
(defun println (obj &optional (stream *standard-output*))
  (let ((*read-default-float-format* 'double-float))
    (prog1 (princ obj stream) (terpri stream))))

(defconstant +mod+ 1000000007)

;;;
;;; Body
;;;

(declaim (inline string16=))
(defun string16= (s1 s2)
  (declare #.OPT
           ((simple-base-string 16) s1 s2))
  (and (= (sb-kernel:%vector-raw-bits s1 0) (sb-kernel:%vector-raw-bits s2 0))
       (= (sb-kernel:%vector-raw-bits s1 1) (sb-kernel:%vector-raw-bits s2 1))))

(declaim (inline sxhash-string16))
(defun sxhash-string16 (string)
  (declare #.OPT
           ((simple-base-string 16) string))
  (macrolet ((update (form)
               `(setq result (ldb (byte 64 0) ,form))))
    (let ((result 0))
      (declare (uint64 result))
      (update (+ result (sb-kernel:%vector-raw-bits string 0)))
      (update (+ result (ash result 10)))
      (update (logxor result (ash result -6)))
      (update (+ result (sb-kernel:%vector-raw-bits string 1)))
      (update (+ result (ash result 10)))
      (update (logxor result (ash result -6)))
      (update (+ result (ash result 3)))
      (update (logxor result (ash result -11)))
      (update (logxor result (ash result 15)))
      (logand result most-positive-fixnum))))

(sb-ext:define-hash-table-test string16= sxhash-string16)

(declaim (inline write-line*))
(defun write-line* (s)
  (declare ((simple-base-string 16) s))
  (loop for c across s
        until (char= c #\Space)
        do #+swank (write-char c)
           #-swank (write-byte (char-code c) *standard-output*))
  (terpri))

(defun main ()
  (declare #.OPT
           (inline sort))
  (let* ((n (read))
         (table (make-hash-table :size n :test 'string16=))
         (max 0))
    (declare (uint62 n max))
    (dotimes (_ n)
      (let ((s (make-string 16 :element-type 'base-char)))
        (read-line-into s)
        (if (gethash s table)
            (incf (the uint32 (gethash s table)))
            (setf (gethash s table) 1))
        (setq max (max max (the uint32 (gethash s table))))))
    (let (res)
      (maphash (lambda (s count)
                 (when (= (the uint32 count) max)
                   (push s res)))
               table)
      (setq res (sort res
                      (lambda (s1 s2)
                        (declare (simple-base-string s1 s2))
                        (string< s1 s2))))
      (dolist (s res)
        (write-line* s)))))

#-swank (main)

;;;
;;; Test and benchmark
;;;

#+swank
(defun io-equal (in-string out-string &key (function #'main) (test #'equal))
  "Passes IN-STRING to *STANDARD-INPUT*, executes FUNCTION, and returns true if
the string output to *STANDARD-OUTPUT* is equal to OUT-STRING."
  (labels ((ensure-last-lf (s)
             (if (eql (uiop:last-char s) #\Linefeed)
                 s
                 (uiop:strcat s uiop:+lf+))))
    (funcall test
             (ensure-last-lf out-string)
             (with-output-to-string (out)
               (let ((*standard-output* out))
                 (with-input-from-string (*standard-input* (ensure-last-lf in-string))
                   (funcall function)))))))

#+swank
(defun get-clipbrd ()
  (with-output-to-string (out)
    (run-program "powershell.exe" '("-Command" "Get-Clipboard") :output out :search t)))

#+swank (defparameter *this-pathname* (uiop:current-lisp-file-pathname))
#+swank (defparameter *dat-pathname* (uiop:merge-pathnames* "test.dat" *this-pathname*))

#+swank
(defun run (&optional thing (out *standard-output*))
  "THING := null | string | symbol | pathname

null: run #'MAIN using the text on clipboard as input.
string: run #'MAIN using the string as input.
symbol: alias of FIVEAM:RUN!.
pathname: run #'MAIN using the text file as input."
  (let ((*standard-output* out))
    (etypecase thing
      (null
       (with-input-from-string (*standard-input* (delete #\Return (get-clipbrd)))
         (main)))
      (string
       (with-input-from-string (*standard-input* (delete #\Return thing))
         (main)))
      (symbol (5am:run! thing))
      (pathname
       (with-open-file (*standard-input* thing)
         (main))))))

#+swank
(defun gen-dat ()
  (uiop:with-output-file (out *dat-pathname* :if-exists :supersede)
    (format out "")))

#+swank
(defun bench (&optional (out (make-broadcast-stream)))
  (time (run *dat-pathname* out)))

;; To run: (5am:run! :sample)
#+swank
(it.bese.fiveam:test :sample
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "7
beat
vet
beet
bed
vet
bet
beet
"
    "beet
vet
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "8
buffalo
buffalo
buffalo
buffalo
buffalo
buffalo
buffalo
buffalo
"
    "buffalo
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "7
bass
bass
kick
kick
bass
kick
kick
"
    "kick
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "4
ushi
tapu
nichia
kun
"
    "kun
nichia
tapu
ushi
")))


Measured memory_bytes: 49512